
Introduction
Cully Technologies, LLC has been in existence since 2000.  He has concentrated on building web sites, web applications, and fat client applications in a variety of commercial industries.  

Kevin has been programming in FoxPro since FoxPro 2.0 for DOS in 1992 and has been programming using the Web Connection framework since 1997.
He utilizes Visual Interdev, Photoshop, Flash, ASP, VBScript, JavaScript, Access, Visual FoxPro, SQL Server, MySQL, and Web Connection.
Kevin enjoys giving presentation to the AFUG in his spare time.

Assumptions

This presentation assumes that you have a full understanding of VFP and object oriented programming.  It is also assumed that you have a foundational understanding of HTML, HTML form variables, and MS IIS.

The Intention of this course

The intention is that you will see, in a very practical way, how to build a web application using VFP as the business and data access layers, HTML as the UI, and VFP tables or SQL Server as the database.  You should walk out of the course having seen:

1. WC installed and tested

2. A web application framework built using the Console.exe program

3. The subclassing (of some) of the WC classes
4. How to query data and return it in a variety of formats

5. How to add, update, and delete data

6. The beginning of securing our web application

What Is Web Connection 

Web connection is a framework.  This framework allows you to leverage your knowledge in VFP to build a fast, lightweight, scalable applications.  Notice that I didn’t say web applications.  I didn’t say it because you can use the WC framework to build fat client applications as well as web applications.

If you have ever worked with ASP (Active Server Pages), WC makes it easy for you to migrate to a more scalable technology by providing a similar (but more powerful) REQUEST and RESPONSE classes.  WC can operate in a scripting mode (similar to ASP) or in compiled mode which you will be seeing today.

WC also provided you with other internet related (buzzword compliant) classes that allow for you to immediately get up to speed with FTP, Email (POP3), XML, SOAP, and Web Services.

The full VFP source code is provided in WC so you can see the actual object hierarchy.  Code, of course, is the ultimate documentation.

In addition to having two ways of interfacing with the UI (scripting and compiled), there are two ways for our WC application to speak with IIS: file based and COM based.  Today we’ll operate in file based mode.  Unfortunately we don’t have time to demonstrate the differences.  COM mode gives a bit of performance boost, robustness, and maintainability for a production environment.  File based mode is easier to get started with, is easier to debug, and has less OS issues.

How IIS and Web Connection talk to each other 

I like to explain how IIS and Web Connection talks to each other as describing them as “Worlds”.  How do we get the IIS world to speak to the Web Connection world?  This is just like IIS is from Mars and WC is from Venus.
Perhaps we need to take one small step back and ask “Why is it a good thing that there’s two worlds at all, instead of just one?”  With simple ASP, there is just one world.  The ASP pages are interpreted inside of IIS.  The ASP pages themselves before interpretation are housed in the IIS space or at least somewhere that IIS has access to them.  If something goes wrong, if the programmer gets sloppy in their code or includes, or (this’ll never happen) if there is a security hole in IIS, it’s possible for your database and system to be exposed to some punk 12 year old malevolent hacker from somewhere in East Jersey or the Philippines or some other such place.  We don’t want that, do we?

With IIS and WC being in two “worlds”, the likelihood of something really bad happening is greatly reduced.  Typically the channels between the worlds are severed but your system hasn’t been compromised.

Let’s take a look at two representations of operating WC: File based and COM.  In the file based document, I’ll show a bit more detail because that’s the mode that we’ll be looking at today.
WC Configuration: COM Based Operation
We’re going to breeze over this very quickly.  Below is a very simple diagram representing your file server.  Installed upon it is IIS, the WC.DLL and the WC Applications registered as COM objects.  Notice that the WC.DLL has control over the instances.  There is a direct pipeline between them.  If the WC.DLL detects that one of the instances has stopped working, it can shut it down and instantiate another instance.  It can also control which instance gets what request.

There are tons of features and issues to go into in regards to COM based operation.  For more information, refer to the WC documentation, Rick Strahl’s book “Internet Applications with Visual FoxPro” or “WebRAD: Building Database Websites with Visual FoxPro and Web Connection” by Harold Chattaway, Randy Pearson, Whil Hentzen.

WC Configuration: File Based Operation
With File based operation, the communication between the WC.DLL and the applications is a very loose one.  A temporary directory is involved and the two are trusting that the other is operating properly to fulfill the request and the response.

When a request is received through IIS and passed to WC.DLL, the WC.DLL creates a small file containing all of the request information and places it into a temporary directory.  The WC applications poll this temporary directory at periodic intervals.  If the WC application detects that there is a request file, it grabs it and begins operating on it.  The other WC instances, go on polling the temporary folder for more requests.
The WC application that has grabbed ahold of the request, operates on it and creates a response file into the temporary folder.  All the time the WC.dll has been polling the temporary folder for any responses.  When it sees the response, it passes it back to IIS which returns it to the requestor (browser).
 SHAPE  \* MERGEFORMAT 



Installing Web Connection 


The installation of WC on your server or personal computer is relatively straightforward.  Once you’ve purchased WC from West-Wind, you will receive instructions on where to download the compressed program files.  Double click on the file and unzip them into the directory that you would like to develop your web applications out of.   

One note of caution:  I recommend C:\webconnect as opposed to C:\wconnect.  The reasoning for this is that the WC setup program will create a directory under IIS named wconnect (c:\inetpub\wwwroot\wconnect).  By having the WC application directory named as something other than ‘wconnect’, it should avoid some confusion in the beginning as to what is executing where.  More on this a bit later.

The unzipping program will attempt to run the SETUP.EXE program for you.  Captured below are some of the screens that you’ll see.

The Web Connection Demo Applications 

Once installed, the program will attempt to start up the WC demo application and pull up a browser pointed to the demo applications.  If you’re wanting to do something over the internet, chances are that the foundation for any task could be found within these demo applications.  Play around here, check out the features of WC, and gather ideas straight from the code samples.

Creating Our Sample Application 

Once again, WC makes it easy to start a new application with a “builder” of sorts that creates a VFP project, subclasses the process class for us.  This “builder” will even setup a script map for our application to mask that we are running WC.  Want to be a Javascript programmer?  Cold Fusion?  PHP?  This is the place to expand your resume!  (kinda.)
From within VFP, at the command prompt, type DO CONSOLE.EXE.  The following screen appears.

Look at all of these buzz-word compliant features!  Can you believe that this comes in a package that costs $399?!?
We’re going to click on the “Create New Project” link at the top to create our application, AFUG Member List.

In step 4, we click on the “Finish” button.  A ton of things are happening at this point.  Subdirectories are created in the IIS area, files are copied into it, a scriptmap is created, the wwProcess class is subclassed into the PRG that becomes the major framework of our application, a browser is opened with the test HTML that contains links to our application, and an attempt is made to build our application.  This normally fails for me but I’ll just open the project, build it and run it.  I’ll click on the links on the HTML page to test out some “Hello World” type functionality.  

Viola!  We have a global, cross platform, OO, high performance, internet application.  Can you feel your self worth increasing?

Just a little Application Cleanup

WC creates our application for us and it runs beautifully.  Our problem is that it has been created in the C:\webconnect folder.  When you’re a programmer for a company, and you are dealing with just a handful of applications, this might not present a problem.  As a contractor, however, once I have reached a dozen applications, my C:\webconnect folder is going to be a mess.  

Let us move some files down into the already created AFUG folder.  Anything that begins with MEMBERLISTAPP.* goes into the AFUG folder.  Anything related to this project that is a PRG or FXP goes into the PRG folder under AFUG.  This includes bld_memberlistapp.prg.  We’ll need a few more files in the AFUG folder to make everything work out as planned: wwIPStuff.dll, wconnect.h, and wconnect_override.h.  

I like to add the wconnect.h and wconnect_override.h into my projects for easy access but exclude them.  We’re going to add an entry into the wconnect_override.h file and uncomment out two other settings.  Make your settings look like this:

#UNDEFINE DEBUGMODE

#DEFINE DEBUGMODE 



.T.

#UNDEFINE SERVER_IN_DESKTOP

#DEFINE SERVER_IN_DESKTOP


.F.

#UNDEFINE WWC_FRAMEWORK_PATH

#DEFINE WWC_FRAMEWORK_PATH


"..\"

Rebuild the project to re-align all of the files to their new location.  Re test your “Hello World” links.  You might have to re-align the temp file settings and the startup settings.  This can be done by clicking on the “Setup” button on the WC application that we’re running.
Invariably someone won’t get things talking after this reorganization.  Remember some of the notes earlier about the two “Worlds” talking together.  To solve your communications problems, follow the communication path:  where is the IIS request file being put?  Is this the same location that the application is looking to pick it up?  Is the application being hit but the response not being sent back to the requestor?  Answer these questions and I think you’ll get file based operations running again in short order.
Subclassing the WC Classes

A major mistake that many programmers make is to not take advantage of the OOP features available through VFP.  This is what separates us from the hoards of VB programmers and we should take advantage of it.  Many of us don’t think about this until it is too late, and then we’d have to go back and retro-fit our inheritance after the fact.

WC takes advantage of OOP in spades while, at the same time, being very light weight and performance oriented.

We’re going to subclass out the wwBusiness class today.  Other classes that you may want to subclass include the wwProcess class, wwShowCursor, among others.

Here’s a diagram of how we’re going to be subclassing the wwBusiness class.  


[image: image12]
Now that we have created work (business) classes for the member.dbf and phones.dbf tables, all we need to do is to set a couple of properties on the classes and we’re almost completely done with a standard way of working with these tables.

Next, we need to tell our web application where to find our new classes and how to instantiate them.

In the definition of our MemberListAppServer, we include our custom properties which will receive our work classes as objects.

*** Add any custom properties here

*** These can act as 'global' vars 

oMember = .NULL.

oPhones = .NULL.

In the SetServerProperties method of our server class, we actually tell the class where to find our other classes and to instantiate them.
*** Add any SET CLASSLIB or SET PROCEDURE code here

SET CLASSLIB TO AFUGBusiness ADDITIVE

SET CLASSLIB TO AFUG_MemberList ADDITIVE

THIS.oMember = CREATEOBJECT("AFUG_Member")

THIS.oPhones = CREATEOBJECT("AFUG_Phones")

ENDFUNC

* SetServerProperties

Querying Data
We will be leveraging our oMember work layer extensively.  Basically we pull the form variables from the HTML form and set them into the object loRequestVars, and then pass them to our custom method: oMember.QueryByParameters().  With the resulting cursor, we decide on how to send it back to the user, HTML, Excel, XML, or PDF.

*********************************************************************

FUNCTION QueryMembers


PRIVATE loRequestVars AS Object


WITH goWCServer.oMember



.GetBlankRecord()



loRequestVars = Request.FormVarsToObject(.oData)



lcOrder = Request.Form("txtOrder")



lcOutput = UPPER(Request.Form("cboOutput"))



.QueryByParameters('c_Member', loRequestVars.m_FName, loRequestVars.m_LName, lcOrder)



DO CASE




CASE lcOutput = "HTML"





THIS.OutputAsHTML('c_Member')




CASE lcOutput = "EXCEL"





THIS.OutputAsEXCEL('c_Member')




CASE lcOutput = "XML"





THIS.OutputAsXML('c_Member')




CASE lcOutput = "PDF"





THIS.OutputAsPDF('c_Member')




OTHERWISE 





THIS.OutputAsHTML('c_Member')



ENDCASE


ENDWITH

ENDFUNC

Lets take a peek inside our oMethod.QueryByParameters methods:

LPARAMETERS tcTargetCursor, tcFName, tcLName, tcOrder

LOCAL lcSelect AS String, lnRetVal AS Integer

lcSelect = [Pk, m_FName, m_LName, m_Joined, m_LastUpdated FROM ] + THIS.cAlias + [ ]

IF EMPTY(tcFName + tcLName)


lcSelect = lcSelect + [WHERE .F. ]

ELSE


lcSelect = lcSelect + [WHERE .T. ] + ;


IIF(EMPTY(tcFName), [], [AND UPPER(m_FName) LIKE '] + ALLTRIM(UPPER(tcFName)) + [%' ]) + ;


IIF(EMPTY(tcLName), [], [AND UPPER(m_LName) LIKE '] + ALLTRIM(UPPER(tcLName)) + [%' ])

ENDIF

lcSelect = lcSelect + [ORDER BY ] + IIF(EMPTY(tcOrder), [m_LName], tcOrder)

lnRetVal = THIS.Query(lcSelect, tcTargetCursor)

IF USED(lcSelect)


SELECT (lcSelect)

ENDIF

RETURN lnRetVal

Displaying As HTML
If we just wanted to display our cursor, we would use wwShowCursor.ShowCursor().  We’re going to take advantage of some built in functionality and use the wwShowCursor.EditRecord() method instead.  We’re halfway to CRUD!  (Create, Read, Update, and Delete.)
********************************************************************

FUNCTION OutputAsHTML

LPARAMETERS tcCursor


IF USED(tcCursor)



loShowCursor = CREATEOBJECT("wwShowCursor")



loShowCursor.cBaseURL = "MemberEdit.afug?"



loShowCursor.cKeyField = "PK"



loShowCursor.cKeyType = "N"



loShowCursor.lAllowAdd = .T.



loShowCursor.lAllowDelete = .T.



loShowCursor.lAlternateRows = .T.



SELECT (tcCursor)



IF loShowCursor.EditTable()




lcCursor = loShowCursor.GetOutput()




lcTemplate = FILE2Var('HTML\MemberList.htm')




lcExpandedHTML = Response.ExpandTemplate(lcTemplate,[],.t.,.t.)




Response.Write(lcExpandedHTML)



ELSE




THIS.StandardPage("Could not create list from cursor.", PROGRAM())



ENDIF


ENDIF

ENDFUNC

Delivering as an XML document
Delivering our data as XML is actually the easiest thing to do!  Easier than returning an HTML table, Excel, or PDF!  Just turn the cursor into an XML string, and then send it back to the requestor!
********************************************************************

FUNCTION OutputAsXML

LPARAMETERS tcCursor


IF USED(tcCursor)



loXML = CREATEOBJECT("wwXML")



loXML.nCreateDataStructure = 1 
&& Schema



lcXML = loXML.CursorToXML()



response.write(lcXML)


ELSE



THIS.StandardPage("Could not find cursor.", PROGRAM())


ENDIF

ENDFUNC

Of course, we are dealing with a simple data structure here.  To create XML with a hierarchical structure, more processing is involved.

Delivering as an Excel (XLS) document

To deliver an Excel document, we’ve got to copy our data to an XLS file but copy it into the IIS “World” where it will be served up.  This is only slightly tricky where we have to keep track of two paths.  (1) the full file path for our application to create the XLS document and (2) the relative path that the browser will use to pull up the file once it’s copied out.
********************************************************************

FUNCTION OutputAsExcel

LPARAMETERS tcCursor


IF USED(tcCursor)



lcRelativeFile = [output\] + DTOC(DATE(),1) + SYS(2015) + [.XLS]



lcOutputFile = goWCServer.oConfig.oMLProcess.cHTMLPagePath + lcRelativeFile



COPY TO (lcOutputFile) TYPE XL5



THIS.StandardPage("Here is your output", [<A href="] + lcRelativeFile + [">] + lcRelativeFile + [</a>])


ELSE



THIS.StandardPage("Could not find cursor.", PROGRAM())


ENDIF

ENDFUNC

Have you noticed anything missing?  When does the XLS file get deleted off of the server?  We’re creating many, many of these files and they would stack up with any volume.  I’ll leave it up to you to come up with your own cleanup procedure.  Here’s some ideas however:
· You could clean up any file in the directory that is older than “X” hours old at the end of each OutputAsExcel hit.

· You could create your own scheduled task that deletes the old files once a day or once and hour.
· And many more.  There are as many solutions and your imagination can come up with.

Delivering as a PDF document

To create PDF documents, you have to have Adobe Distiller, ActivePDF, Amyumi or some other PDF creation program installed on the server that will create the PDFs.  Adobe has recently changed their licensing requirements to check them carefully to make sure that you are in compliance.  The approach we’re taking is similar to the Excel creation in that we place the PDF document into the IIS “World” to be served up.  We provide a page with a link on it for the user to click on and retrieve their document.  Again, cleanup is an issue and one that I haven’t addressed here.

********************************************************************

FUNCTION OutputAsPDF

LPARAMETERS tcCursor


IF USED(tcCursor)



lcRelativeFile = [output\] + DTOC(DATE(),1) + SYS(2015) + [.PDF]



lcOutputFile = goWCServer.oConfig.oMLProcess.cHTMLPagePath + lcRelativeFile



loPDF = CREATEOBJECT("wwPDF40")



loPDF.PrintReport("rpt\rptMemberList.Frx",lcOutputFile)



IF EMPTY(loPDF.cErrorMsg)




THIS.StandardPage("Here is your output", [<A href="] + lcRelativeFile + [">] + lcRelativeFile + [</a>])



ELSE




THIS.ErrorMsg(loPDF.cErrorMsg)



ENDIF


ELSE



THIS.StandardPage("Could not find cursor.", PROGRAM())


ENDIF

ENDFUNC

CRUD - Create, Retrieve, Update, Delete a single record

The work layer is our friend and makes this task almost trivial.  Structuring the logic can be a little tricky, but once you’ve got it right, the template will get you far.  The code you see below is an over simplification.  You’ll want to do plenty of data validation and error checking.
*********************************************************************

FUNCTION MemberEdit


lcAction = UPPER(Request.QueryString("Action"))


IF EMPTY(lcAction)



lcAction = UPPER(Request.Form("btnAction"))


ENDIF


lnMemberPK = INT(VAL(ALLTRIM(Request.QueryString("ID"))))


DO CASE



CASE lcAction = "QUERY"




THIS.QueryMembers()



CASE lcAction = "DELETE"




goWCServer.oMember.oData.PK = lnMemberPK




goWCServer.oMember.DELETE()




THIS.StandardPage("The member record was deleted.", ;

[<a href="QueryMembers.afug">Return to Query Page</a>])



OTHERWISE




DO CASE





CASE lcAction = "ADD"






lnMemberPK = goWCServer.oMember.CreateNewID()






goWCServer.oMember.QueryByParameters('c_Member',[],[],[])






lcMessage = [New Record]





OTHERWISE






IF goWCServer.oMember.Load(lnMemberPK)







IF lcAction = "SAVE"








goWCServer.oMember.oData = ;

 Request.FormVarsToObject(goWCServer.oMember.oData)








goWCServer.oMember.Save()







    lcMessage=[Record Saved]







ELSE








lcMessage = []







ENDIF






ELSE






   THIS.ErrorMsg("No Customer Number to edit", ;

"Please enter a customer ID to edit")






ENDIF




ENDCASE




*** Display




loSC = CREATE("wwShowCursor")




loSC.EditRecord()




lcTable = loSC.GetOutput()




lcTemplate = File2Var('HTML\MemberEdit.htm')




lcExpandedHTML = Response.ExpandTemplate(lcTemplate, [], .T., .T.)




Response.Write(lcExpandedHTML)


ENDCASE

ENDFUNC

Security - Logging in Users

Web Connection has security “built in” but commented out.  This may seem strange at first but the wwProcess class, in the Process method calls to the wwProcess.Login method but is commented out.  If you were to uncomment this, you could immediately take advantage of Basic Authentication.  Basic Authentication takes advantage of NT based security where each user that logs in must be a user set up on the server.  This is a lot of work, in my opinion.  I don’t want to have to go through NT security to add a user every time I want to grant access to someone for my web application.  Have you checked how much a per user license is now days?
I prefer a database based security.  This makes sense doesn’t it?  We’re database programmers, we can easily check a User ID and Password against a database no problem!  In fact our Members.DBF is its own login database.  We’ll validate the User ID and Password against the corresponding fields in Member.DBF.

Cookies.  The word strikes fear into the hearts of millions!  Hard to believe.  I know that the person that coined this word feels like … well, he should feel terrible.  This probably set the internet back two years because of the connotations of this word.  Ah well.  That’s the way the cookie crumbles.  (Had to do it!  Sorry!)
Cookies are extremely useful to us and aren’t abused like they used to be.  Cookies are a small file that is passed back and forth from server to browser typically when displaying a login screen, and then returned to the server on each subsequent hit.  Nothing should be placed into the cookie except for a unique session id, with few exceptions.  Cookies were abused early on when people passed actual user IDs and passwords back and forth in cookies.  Web Connection uses a SYS(2015) to create their session ID which is the only thing we place into our cookie.  We can look up this session ID in our … yes, session table, and find out all we need to know about this user.  

Unfortunately I don’t have time to show some of the advanced things you can do with sessions but just let me say that wwSession.SetSessionVar and wwSession.GetSessionVar are your friends.  We’re just going to validate that the user has a valid session after a login.  This also checks that the browser has not “timed out”.  If the user hasn’t had any activity on the session record after “X” seconds (default of 1,800 seconds or ½ and hour), then the session is no longer valid, and we’ll force them to log in again.  This prevents people from sitting down at a desk after someone has gone to lunch and having this person view sensitive information.

Enough talk, let’s see some code!

First, we override the wwProcess.Process method in our class.  Notice our GetCookie method, our wwSession.IsValidSession method, and our call to THIS.Login().  These calls are key.

*********************************************************************

FUNCTION Process


THIS.cSessionID = THIS.oRequest.GetCookie(This.cAppName)


loSession = CREATEOBJECT("wwSession")


IF loSession.IsValidSession(THIS.cSessionID) 


    DODEFAULT()


ELSE



loSession.EndSession(THIS.cSessionID)



IF NOT THIS.Login()



   RETURN .F.



ENDIF


ENDIF

RETURN .T.

ENDFUNC

Next we’ll look at our Login method.  This is a bit more complex than bare bones because I’m determining which field should have the SetFocus if the login is rejected and we have to show the login page again.

*********************************************************************

FUNCTION Login

PARAMETER pcMessage, lcUserID, lcPWD

LOCAL lcCidClient


Request = goWCServer.oRequest


Response = THIS.oResponse


pcTitle = goWCServer.cAppName


IF VARTYPE(pcMessage)#"C"



RELEASE pcMessage



pcMessage = []


ENDIF


IF VARTYPE(lcUserID) = "C" AND VARTYPE(lcUserPWD) = "C" 



* The values were passed in, lets use them.



lcUserID 
= ALLTRIM(UPPER(lcUserid))



lcPWD 
= ALLTRIM(UPPER(lcPWD))


ELSE



lcProcedure = UPPER(Request.Querystring(2))



IF lcProcedure # [LOGIN]




* The users session most likely timed out.  Don't use additl parameters. - KJC




STORE [] TO lcUserID, lcPWD



ENDIF


ENDIF


IF EMPTY(lcUserID) AND EMPTY(lcPWD)



lcUserID = ALLTRIM(UPPER(Request.Form([txtuserid])))



lcPWD 
= ALLTRIM(UPPER(Request.Form([txtpwd])))


ENDIF


IF VARTYPE(lcUserID) = "C" AND VARTYPE(lcPWD) = "C"



*** If the userid and pwd is passed in, then go ahead and check for valid user.



*** This site, if using this method, should definitely use HTTPS. - KJC



IF NOT EMPTY(lcPWD)




IF goWCServer.oMember.QueryUserByIDAndPassword('c_User', lcUserID, lcPWD) > 0





loSession.NewSession(lcUserID, THIS.cSessionID)





THIS.QueryMembers()





RETURN .T.




ENDIF



ENDIF


ENDIF


*** This deals with the ability to set focus.


DO CASE



CASE EMPTY(lcPWD)




lcFocus = [txtpwd]



OTHERWISE




lcFocus = [txtuserid]


ENDCASE



lcSessionID=SUBSTR(SYS(2015),2,9)


loHeader=CREATE("wwHTTPHeader")


loHeader.Defaultheader()


loHeader.AddCookie(This.cAppName, lcSessionID)


Response.clear()


Response.ContentTypeHeader(loHeader)


lcTemplate = File2Var("HTML\Login.HTM")


lcExpandedhtml = Response.ExpandTemplate(lcTemplate,[],.t.,.t.)


response.write(lcExpandedHTML)

RETURN .F.

ENDFUNC

Conclusion
If you are reading this, then you’re still awake.  This is good.
As I was putting this together, I was torn about what to include and what to exclude.  I think that I had set a very aggressive schedule to show as many features as I possible could within a “Web Connection Crash Course 101” type class.  Some of the things that I’ve shown you, I don’t actually use.  wwShowCursor.EditRecord is one of them.  I like more control over my own HTML so I normally hand code for the display of the information.  I decided to talk about subclassing the Web Connection classes and I approached this project in a 4 tier approach which I love for people to strive for.  These two issues are what raise us above the other “Hack” programmers.  (We’ve worked with many of those, eh?)

There are many other topics I would never even think to include in a course like this but are extremely important, some are basic and some are advanced.  Email, Asynchronous processing, Charting and Graphing, Soap and Web Services.

Then there are the other peripheral web issues, outside of Web Connection, such as load balancing, performance and stress testing, DBMS issues, scalability, browser compatibility, and numerous others.

Rick Strahl and numerous other contributors have made Web Connection the best web development tool in existence, in my opinion.  I don’t know how many times that I’ve thought that “Web Connection makes me look like a genius.”
Thanks for your time and interest.

Now get out there and write a killer web app!

